22.13 Die Makrosprache DIESEL 

Im Menü können Sie Ausdrücke der Makrosprache DIESEL verwenden, um Abrollmenüs je nach Programmzustand zu verändern bzw. Werte oder Zeichenketten an AutoCAD-LT-Befehle zu übergeben. DIESEL ist die Abkürzung für Direct Interpretively Evaluation String Expression Language und Ausdrücke von DIESEL beginnen immer mit dem Zeichen $. 

	
	Aktion: DIESEL zur Ausblendung von Menüfunktionen 


Menüfunktionen lassen sich je nach Zustand des Programms ausblenden. Die entsprechende Funktion erscheint dann im Menü grau gerastert und kann nicht angewählt werden. Dazu muss in der Menüdatei vor die Bezeichnung »~« gesetzt werden: 

[~&Mansfen]^C^CMansfen                Funktion ist ausgeblendet

[~->&Mansfen]                Untermenü ist ausgeblendet

Ein Menüteil, der immer ausgeblendet ist, ergibt jedoch wenig Sinn. Deshalb lassen sich Menüteile abhängig von bestimmten Zuständen ausblenden. Diese Abhängigkeiten werden mit DIESEL-Ausdrükken angegeben. Im obigen Beispiel würde man den Befehl Mansfen ausblenden, wenn die Systemvariable Tilemode den Wert 1 hat. Die DIESEL-Funktion Getvar liefert den Wert einer Systemvariablen und die Funktion If ermöglicht bedingte Ausführungen (siehe unten). Das könnte dann so aussehen: 

[$(if,$(getvar,tilemode),~)&Mansfen]^C^CMansfen

oder 

[->$(if,$(getvar,tilemode),~)&Mansfen]

	Aktion: DIESEL zur Markierung von Menüfunktionen 


Menüfunktionen können mit einer Markierung versehen werden, einem Häkchen, das vor die entsprechende Funktion gesetzt ist. Damit kann angezeigt werden, dass eine Funktion bereits gesetzt ist. Vor den Anzeigetext muss die Zeichenkombination »!.« gesetzt werden: 

[!.Ortho]^O            Funktion ist markiert

[!.Fang]^B            Funktion ist markiert

Auch das ist nur sinnvoll, wenn es vom Programmzustand abhängig gemacht wird. Im Beispiel oben ist es sinnvoll, die Funktionen zu markieren, wenn sie bereits eingeschaltet sind. Die Zeichenmodi sind in Systemvariablen gespeichert, deren Zustand mit einem DIESEL-Makro abgefragt werden kann: 

[$(if,$(getvar,orthomode),!.)&Ortho]^O

[$(if,$(getvar,snapmode),!.)&Fang]^B

Beide Funktionen können auch kombiniert werden. So würde im folgenden Beispiel der Befehl markiert und grau gerastert werden, wenn die Systemvariable Tilemode den Wert 1 hat: 

[$(if,$(getvar,tilemode),~!.)Mansfen]^C^CMansfen

	Aktion: Wechselnde Anzeigen gesteuert von DIESEL 


Sollen Funktionen in Menüdateien je nach Programmzustand wechseln, so kann das ebenfalls über DIESEL-Ausdrücke gesteuert werden. Soll beispielsweise eine andere Anzeige erscheinen, je nachdem, ob der Ortho-Modus aus- oder eingeschaltet ist: 

[$(if,$(getvar,orthomode),Ortho &aus,Ortho &ein)]^O

	Aktion: DIESEL innerhalb der Anzeige oder der Menüfunktion 


Innerhalb der Anzeige oder der Menüfunktion lassen sich DIESEL-Ausdrücke zur Berechnung von Werten oder zur Ausgabe von Texten benützen. Die Ergebnisse lassen sich anzeigen oder an AutoCAD-LT-Befehle übergeben. 

Erscheinen DIESEL-Ausdrücke in einer Menüfunktion, haben sie die Form $M=$(Ausdruck). Im Folgenden einige Beispiele: 

Anzeige des aktuellen Layers im Menü 

[$(eval,"Aktueller Layer: "$(getvar,clayer))]

Anzeige des aktuellen Datums im Menü 

[$(edtime,$(getvar,date),DDD", "D MON YYYY)]

Bedingte Ausführung von Befehlen 

[->/WWahl]

  [&Letztes]$M=$(if,$(getvar,cmdactive),,_select;)_last

  [&Alle]$M=$(if,$(getvar,cmdactive),,_select;)_all

  [&Zaun]$M=$(if,$(getvar,cmdactive),,_select;)_fence

Die Systemvariable Cmdactive ist 1, wenn ein Befehl aktiv ist. Wenn also in den obigen Beispielen ein Editierbefehl aktiv ist, wird _select (Befehl Wahl] nicht ausgeführt, sondern nur die Option hinter der Klammer. 

Abfrage von Werten 

[Objekt auf &akt.Layer]^C^C_change \ EI LAYER;$M=$(getvar,clayer);;

Die gewählten Objekte werden mit dem Befehl Ändern auf den aktuellen Layer gebracht. 

Berechnen von Werten 

[Zoom 30%]^C^CZOOM $M=$(/,1,0.7)XP

[Zoom 50%]^C^CZOOM $M=$(/,1,0.5)XP

Ein Fenster im Papierbereich wird um einen berechneten Faktor vergrößert. 

22.14 DIESEL-Funktionen 

DIESEL-Funktionen eignen sich für Statusabfragen, Berechnungen und Anzeige-Funktionen. In den folgenden Übersichten finden Sie alle Funktionen aufgelistet. Alle Funktionen sind auf 10 Parameter begrenzt, einschließlich des Funktionsnamens. 

Berechnung: Der Wert der Berechnung wird von der Funktion zurückgegeben. 

Addition          $(+,Wert1{,Wert2,...Wert9})

Subtraktion          $(-,Wert1{,Wert2,...Wert9})

Multiplikation          $(*,Wert1{,Wert2,...Wert9})

Division          $(/,Wert1{,Wert2,...Wert9})

Vergleich: Trifft die Bedingung zu, liefert die Funktion 1, ansonsten 0. 

Gleichheit          $(=,Wert1,Wert2)

Kleiner als          $(<,Wert1,Wert2)

Größer als          $(>,Wert1,Wert2)

Ungleichheit          $(!=,Wert1,Wert2)

Kleiner gleich          $(<=,Wert1,Wert2)

Größer gleich        $(>=,Wert1,Wert2)

Logische Verknüpfung: Die Funktion liefert die bitweise logische Verknüpfung von Ganzzahlen. 

AND      $(and,Wert1{Wert2,...Wert9})

OR      $(or,Wert1{Wert2,...Wert9})

XOR      $(xor,Wert1{Wert2,...Wert9})

Winkel umwandeln: Der angegebene Wert wird als Winkel verarbeitet. Das Format kann mit Modus und Genauigkeit angegeben werden. Ohne Angabe werden die Einstellungen des Befehls Einheit übernommen. 

Angtos      $(angtos,Wert{,Modus,Genauigkeit})

Modus      Format

0      Grad

1      Grad/Minuten/Sekunden

2      Neugrad

3      Bogenmaß

4      Feldmaß

Datum: Das Datum, das aus der Systemzeit ausgelesen wird (zum Beispiel aus der Systemvariablen Date), wird entsprechend der Angabe im Muster verarbeitet. Das Muster besteht aus Formatmustern, die durch die entsprechenden Datums- und Zeitangaben ersetzt werden. 

Edtime      $(edtime,Zeit,Muster)

Die folgende Formattabelle zeigt die Ausgabe am Beispiel des Datums 8. Juli 1999, 11.30 Uhr, 12 Sekunden: 

Format      Ausgabe

D      8

DD      08

DDD      Donnerstag

M      7

MO      07

MON      Jul.

MONTH      Juli

YY      99

YYYY      1999

H      11

HH      11

MM      30

SS      12

AM/PM      AM

am/pm      am

A/P      A

a/p      a

Das Beispiel von oben liefert bei folgender Funktion: 

$(edtime,$(getvar,date),DDD"," DD MON YYYY - H:MMam/pm

das Ergebnis: Donnerstag, 08 Jul. 1999 - 11:30am 

Texte vergleichen: Die Funktion gibt 1 zurück, wenn zwei Texte gleich sind. 

EQ      $(eq,Text1,Text2)

Text übergeben: Der Text wird an die DIESEL-Funktion übergeben. 

EVAL      $(eval,Text)

Umwandlung von Zahlen: Aus der reellen Zahl wird eine Ganzzahl gebildet, indem Stellen abgeschnitten werden. 

FIX      $(fix,Zahl)

Abfrage von Variablen: Der Wert der Umgebungsvariablen bzw. der Systemvariablen wird zurückgegeben. 

GETENV      $(getenv,Variable)

GETVAR      $(getvar,Variable)

Bedingte Ausführung: Wenn der Ausdruck ungleich 0 ist, wird Aktion1 ausgewertet, ansonsten Aktion2. 

IF      $(if,Ausdruck,Aktion1{,Aktion2})

Indiziert auswählen: Das Element mit der angegebenen Nummer wird aus dem Text ausgewählt und zurückgegeben. Die Zeichenfolge muss aus Werten bestehen, die durch Kommas eingegrenzt sind. Der erste Wert hat die Nummer 0. Die Funktion bietet sich an, um X-, Y- oder Z-Komponenten von Punkten abzurufen, die von der Funktion getvar zurückgegeben werden. 

INDEX      $(index,Nummer,Text)

Das Element mit der angegebenen Nummer wird aus den verschiedenen Argumenten ausgewählt und zurückgegeben. 

NTH      $(index,nth,Nummer,Wert0 {,Wert1,...,Wert9})

Zahl umwandeln: Der angegebene Wert wird als reelle Zahl editiert. Das Format kann mit Modus und Genauigkeit angegeben werden. Ohne Angabe werden die Einstellungen des Befehls Einheit übernommen. 

RTOS      $(rtos,Wert{,Modus,Genauigkeit})

Zeichenfolgen umwandeln: Die Länge des Textes wird zurückgegeben. 

STRLEN      $(strlen,Textes)

Der Teiltext des Textes wird am Zeichen Start und mit der angegebenen Länge zurückgegeben. 

SUBSTR      $(substr,Teiltext,Start{,Länge})

Der Text wird in Großbuchstaben zurückgegeben. 

UPPER      $(upper,Text)

